HEWLETT-PACKARD

$36

HP-34C

MATHEMATICS Applications

NOTICE

HEWLETT PACKARD

HP-34C

Mathematics Applications

September 1979

00034-90032

Printed in Singapore

Introduction

This Mathematics Applications book was written to help you get the most from your HP-34C calculator. The programs were chosen to provide useful calculations for many of the common problems encountered in mathematics.

They provide you with immediate problem solving capabilities in your everyday work. You may also find them useful as guides to programming techniques for writing your own customized software.

You will find general information on how to key in and run programs under ''A Word About Program Usage'' in the Applications book you received with your calculator.

We hope this Mathematics Applications book will be a valuable tool in your work. We would appreciate any comments you may have about it.

Contents

Introduction 2

Calculus and Linear Systems
System of Linear Equations with Three Unknowns 4
Determinant and Inverse of a 3×3 Matrix 9
Numerical Integration by Discrete Points 14
Differential Equations 18

Geometry
Vector Operations 28
Triangle Solutions 33
Coordinate Transformation 41
Circle Determined by Three Points 46

Other
Complex Operations 50
Base Conversions 59

System of Linear Equations With 3 Unknowns

This program uses Cramer's rule to solve systems of linear equations with three unknowns.

Equations:

A system of linear equations can be expressed as

$$Ax = b$$

For 3 Unknowns, $$A = \begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix}$$

$$x = \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} \qquad b = \begin{bmatrix} b_1 \\ b_2 \\ b_3 \end{bmatrix}$$

Determinant of the system

$$\text{Det} = a_{11}(a_{22}a_{33}-a_{23}a_{32}) - a_{12}(a_{21}a_{33}-a_{23}a_{31}) + a_{13}(a_{21}a_{32}-a_{22}a_{31})$$

b_i's are solved by $b_i = \dfrac{\det(i)}{\text{Det}}$ for Det $\neq 0$

Where det (i) is the determinant of the A matrix with the i_{th} column replaced by b.

Remark:

If Det = 0 then the system is linearly dependent and this program is not applicable. The program will terminate with error 0.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
GSB 0	002– 13 0
STO 0	003– 23 0
R/S	004– 74
h LBL 0	005–25, 13, 0
0	006– 0
STO f I	007–23, 14, 23
RCL 6	008– 24 6
RCL 8	009– 24 8
GSB 9	010– 13 9
RCL 4	011– 24 4
RCL 9	012– 24 9
GSB 9	013– 13 9
RCL 5	014– 24 5
RCL 7	015– 24 7
GSB 9	016– 13 9
CHS	017– 32
RCL 3	018– 24 3
RCL 8	019– 24 8
GSB 9	020– 13 9
RCL 1	021– 24 1
RCL 9	022– 24 9
GSB 9	023– 13 9
RCL 2	024– 24 2
RCL 7	025– 24 7
h LBL 9	026–25, 13, 9
g DSE	027– 15 23
h PSE	028– 25 74

KEY ENTRY	DISPLAY
RCL f (i)	029–24, 14, 24
×	030– 61
×	031– 61
+	032– 51
h RTN	033– 25 12
h LBL B	034–25, 13, 12
STO • 3	035– 23 .3
g R↓	036– 15 22
STO • 2	037– 23 .2
g R↓	038– 15 22
STO • 1	039– 23 .1
R/S	040– 74
h LBL 1	041–25, 13, 1
1	042– 1
GSB 7	043– 13 7
R/S	044– 74
h LBL 2	045–25, 13, 2
4	046– 4
GSB 7	047– 13 7
R/S	048– 74
h LBL 3	049–25, 13, 3
7	050– 7
GSB 7	051– 13 7
R/S	052– 74
h LBL 7	053–25, 13, 7
STO • 4	054– 23 .4
STO f I	055–23, 14, 23
GSB 8	056– 13 8
GSB 0	057– 13 0

KEY ENTRY	DISPLAY
[RCL] 0	058- 24 0
[÷]	059- 71
[STO] [•] 0	060- 23 .0
[RCL] [•] 4	061- 24 .4
[STO] [f] [I]	062-23, 14, 23
[GSB] 8	063- 13 8
[RCL] [•] 0	064- 24 .0
[h] [RTN]	065- 25 12
[h] [LBL] 8	066-25, 13, 8
[RCL] [•] 1	067- 24 .1
[h] [x⇄(i)]	068- 25 21
[STO] [•] 1	069- 23 .1
[g] [ISG]	070- 15 24
[h] [PSE]	071- 25 74
[RCL] [•] 2	072- 24 .2
[h] [x⇄(i)]	073- 25 21
[STO] [•] 2	074- 23 .2
[g] [ISG]	075- 15 24
[h] [PSE]	076- 25 74
[RCL] [•] 3	077- 24 .3
[h] [x⇄(i)]	078- 25 21
[STO] [•] 3	079- 23 .3
[h] [RTN]	080- 25 12

REGISTERS			I Index
R_0 Det	R_1 a_{11}	R_2 a_{21}	R_3 a_{31}
R_4 a_{12}	R_5 a_{22}	R_6 a_{32}	R_7 a_{13}
R_8 a_{23}	R_9 a_{33}	$R_{.0}$ det (i)	$R_{.1}$ b_1
$R_{.2}$ b_2	$R_{.3}$ b_3	$R_{.4}$ Index	$R_{.5}$—$R_{.7}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store elements of A matrix.	a_{11}	[STO] 1	a_{11}
		a_{21}	[STO] 2	a_{21}
		a_{31}	[STO] 3	a_{31}
		a_{12}	[STO] 4	a_{12}
		a_{22}	[STO] 5	a_{22}
		a_{32}	[STO] 6	a_{32}
		a_{13}	[STO] 7	a_{13}
		a_{23}	[STO] 8	a_{23}
		a_{33}	[STO] 9	a_{33}
3	Calculate determinant.		[A]	Det
4	Input b.	b_1	[ENTER↑]	
		b_2	[ENTER↑]	
		b_3	[B]	b_1
5	Calculate x_1.		[GSB] 1	x_1
6	Calculate x_2.		[GSB] 2	x_2
7	Calculate x_3.		[GSB] 3	x_3
8	For a new b with the same			
	system, go to step 4.			
9	For a new system, go to step 2.			

Example:

Find x_1, x_2, and x_3 for the following system.

$$\begin{bmatrix} 19 & -4 & -15 \\ -4 & 22 & -10 \\ -15 & -10 & 26 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} = \begin{bmatrix} 40 \\ 0 \\ 0 \end{bmatrix}$$

Keystrokes:	Display:	
[f] [FIX] 4		
19 [STO] 1		
4 [CHS] [STO] 2		
15 [CHS] [STO] 3		
4 [CHS] [STO] 4		
22 [STO] 5		
10 [CHS] [STO] 6		
15 [CHS] [STO] 7		
10 [CHS] [STO] 8		
26 [STO] 9		
[A]	**2,402.0000**	(Det)
40 [ENTER↑] 0 [ENTER↑]		
0 [B]	**40.0000**	
[GSB] 1	**7.8601**	(x_1)
[GSB] 2	**4.2298**	(x_2)
[GSB] 3	**6.1615**	(x_3)

Determinant and Inverse of a 3 × 3 Matrix

This program calculates the determinant and the inverse of a 3 × 3 matrix (if it exists, i.e., determinant ≠ 0).

Equation:

$$\text{Let } A = \begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix}$$

then

$$A^{-1} = \frac{1}{\det} \begin{bmatrix} \begin{vmatrix} a_{22} & a_{23} \\ a_{32} & a_{33} \end{vmatrix} & -\begin{vmatrix} a_{12} & a_{13} \\ a_{32} & a_{33} \end{vmatrix} & \begin{vmatrix} a_{12} & a_{13} \\ a_{22} & a_{23} \end{vmatrix} \\ -\begin{vmatrix} a_{21} & a_{23} \\ a_{31} & a_{33} \end{vmatrix} & \begin{vmatrix} a_{11} & a_{13} \\ a_{31} & a_{33} \end{vmatrix} & -\begin{vmatrix} a_{11} & a_{13} \\ a_{21} & a_{23} \end{vmatrix} \\ \begin{vmatrix} a_{21} & a_{22} \\ a_{31} & a_{32} \end{vmatrix} & -\begin{vmatrix} a_{11} & a_{12} \\ a_{31} & a_{32} \end{vmatrix} & \begin{vmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{vmatrix} \end{bmatrix}$$

where

$$\det = a_{11}(a_{22}a_{33}-a_{23}a_{32}) - a_{12}(a_{21}a_{33}-a_{23}a_{31}) + a_{13}(a_{12}a_{32}-a_{22}a_{11})$$

and

$$\begin{vmatrix} a & b \\ c & d \end{vmatrix} = ad - bc$$

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-
[h] [LBL] [A]	001-25, 13, 11
0	002- 0
[STO] [f] [I]	003-23, 14, 23
[RCL] 6	004- 24 6
[RCL] 8	005- 24 8
[GSB] 9	006- 13 9
[RCL] 4	007- 24 4
[RCL] 9	008- 24 9
[GSB] 9	009- 13 9
[RCL] 5	010- 24 5
[RCL] 7	011- 24 7
[GSB] 9	012- 13 9
[CHS]	013- 32
[RCL] 3	014- 24 3
[RCL] 8	015- 24 8
[GSB] 9	016- 13 9
[RCL] 1	017- 24 1
[RCL] 9	018- 24 9
[GSB] 9	019- 13 9
[RCL] 2	020- 24 2
[RCL] 7	021- 24 7
[h] [LBL] 9	022-25, 13, 9
[g] [DSE]	023- 15 23
[h] [PSE]	024- 25 74
[RCL] [f] [(i)]	025-24, 14, 24
[×]	026- 61
[×]	027- 61
[+]	028- 51
[STO] 0	029- 23 0
[h] [RTN]	030- 25 12
[h] [LBL] 1	031-25, 13, 1
[RCL] 5	032- 24 5
[RCL] 9	033- 24 9
[RCL] 8	034- 24 8
[RCL] 6	035- 24 6
[GSB] 8	036- 13 8
[R/S]	037- 74
[RCL] 8	038- 24 8
[RCL] 3	039- 24 3
[RCL] 2	040- 24 2
[RCL] 9	041- 24 9
[GSB] 8	042- 13 8
[R/S]	043- 74
[RCL] 2	044- 24 2
[RCL] 6	045- 24 6
[RCL] 5	046- 24 5
[RCL] 3	047- 24 3
[GSB] 8	048- 13 8
[R/S]	049- 74
[h] [LBL] 2	050-25, 13, 2
[RCL] 7	051- 24 7
[RCL] 6	052- 24 6
[RCL] 4	053- 24 4
[RCL] 9	054- 24 9
[GSB] 8	055- 13 8
[R/S]	056- 74
[RCL] 1	057- 24 1

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[RCL] 9	058- 24 9	[RCL] 1	078- 24 1
[RCL] 7	059- 24 7	[RCL] 8	079- 24 8
[RCL] 3	060- 24 3	[GSB] 8	080- 13 8
[GSB] 8	061- 13 8	[R/S]	081- 74
[R/S]	062- 74	[RCL] 1	082- 24 1
[RCL] 4	063- 24 4	[RCL] 5	083- 24 5
[RCL] 3	064- 24 3	[RCL] 4	084- 24 4
[RCL] 1	065- 24 1	[RCL] 2	085- 24 2
[RCL] 6	066- 24 6	[GSB] 8	086- 13 8
[GSB] 8	067- 13 8	[R/S]	087- 74
[R/S]	068- 74	[GTO] 1	088- 22 1
[h] [LBL] 3	069-25, 13, 3	[h] [LBL] 8	089-25, 13, 8
[RCL] 4	070- 24 4	[×]	090- 61
[RCL] 8	071- 24 8	[g] [R↓]	091- 15 22
[RCL] 7	072- 24 7	[×]	092- 61
[RCL] 5	073- 24 5	[f] [R↑]	093- 14 22
[GSB] 8	074- 13 8	[−]	094- 41
[R/S]	075- 74	[RCL] 0	095- 24 0
[RCL] 7	076- 24 7	[÷]	096- 71
[RCL] 2	077- 24 2	[h] [RTN]	097- 25 12

	REGISTERS		I Unused
R_0 Det	R_1 a_{11}	R_2 a_{21}	R_3 a_{31}
R_4 a_{12}	R_5 a_{22}	R_6 a_{32}	R_7 a_{13}
R_8 a_{23}	R_9 a_{33}	$R_{.0}$—$R_{.5}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the matrix columnwise:	a_{11}	[STO] 1	
		a_{21}	[STO] 2	
		a_{31}	[STO] 3	
		a_{12}	[STO] 4	
		a_{22}	[STO] 5	
		a_{32}	[STO] 6	
		a_{13}	[STO] 7	
		a_{23}	[STO] 8	
		a_{33}	[STO] 9	
3	Calculate the determinant.		[A]	Det
	Calculate the			
	Inverse			
4	First column		[GSB] 1	a'_{11}
			[R/S]	a'_{21}
			[R/S]	a'_{31}
5	Second column		[GSB] 2	a'_{12}
			[R/S]	a'_{22}
			[R/S]	a'_{32}
6	Third column		[GSB] 3	a'_{13}
			[R/S]	a'_{23}
			[R/S]	a'_{33}
7	If you want the results again,			
	go to any of the steps 4, 5 or 6.			
8	For a new case, go to step 2.			

Example:

Find the determinant and inverse of the following matrix:

$$\begin{bmatrix} 23 & 15 & 17 \\ 8 & 11 & -6 \\ 4 & 15 & 12 \end{bmatrix}$$

Keystrokes:	Display:	
[f] [FIX] 4		
23 [STO] 1 8 [STO] 2		
4 [STO] 3 15 [STO] 4		
11 [STO] 5 15 [STO] 6		
17 [STO] 7		
6 [CHS] [STO] 8		
12 [STO] 9		
[A]	***4,598.0000***	(det)
[GSB] 1	***0.0483***	(a'_{11})
[R/S]	***-0.0261***	(a'_{21})
[R/S]	***0.0165***	(a'_{31})
[GSB] 2	***0.0163***	(a'_{12})
[R/S]	***0.0452***	(a'_{22})
[R/S]	***-0.0620***	(a'_{32})
[GSB] 3	***-0.0602***	(a'_{13})
[R/S]	***0.0596***	(a'_{23})
[R/S]	***0.0289***	(a'_{33})

Numerical Integration By Discrete Points

This program will perform numerical integration when a function is known at a finite number of equally spaced points (discrete case). The integrals are approximated by either the trapezoidal rule or Simpson's rule.

Equations:

Let $x_0, x_1, \ldots, x_n$ be $n + 1$ equally spaced points ($x_j = x_0 + jh, j = 0, 1, 2, \ldots, n$) at which corresponding values $f(x_0), f(x_1) \ldots, f(x_n)$ of the function f(x) are known.

The integral: $\int_{x_0}^{x_n} f(x)\,dx$ may be approximated using

1.

The trapezoidal rule:

$$\int_{x_0}^{x_n} f(x)\,dx \sim \frac{h}{2}\left[f(x_0) + 2\left(\sum_{j=1}^{n-1} f(x_j)\right) + f(x_n)\right]$$

2.

Simpson's rule:

$$\int_{x_0}^{x_n} f(x)\,dx \sim \frac{h}{3}\Big[f(x_0) + 4f(x_1) + 2f(x_2) +$$

$$\ldots + 4f(x_{n-3}) + 2f(x_{n-2}) + 4f(x_{n-1}) + f(x_n)\Big]$$

In order to apply Simpson's rule, n must be even. If n is not even, "Error 0" will be displayed.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
STO 4	002- 23 4
R/S	003- 74
h LBL B	004-25, 13, 12
STO 0	005- 23 0
STO 5	006- 23 5
0	007- 0
STO 3	008- 23 3
h LBL 9	009-25, 13, 9
R/S	010- 74
h LBL B	011-25, 13, 12
STO 1	012- 23 1
GSB 6	013- 13 6
ENTER↑	014- 31
+	015- 51
STO + 5	016-23, 51, 5
1	017- 1
STO + 3	018-23, 51, 3
RCL 3	019- 24 3
R/S	020- 74
h LBL B	021-25, 13, 12
STO 1	022- 23 1
GSB 6	023- 13 6
STO + 5	024-23, 51, 5
1	025- 1
STO + 3	026-23, 51, 3
RCL 3	027- 24 3
GTO 9	028- 22 9

KEY ENTRY	DISPLAY
h LBL 1	029-25, 13, 1
2	030- 2
RCL 0	031- 24 0
GTO 0	032- 22 0
h LBL 2	033-25, 13, 2
RCL 3	034- 24 3
GSB 7	035- 13 7
3	036- 3
RCL 5	037- 24 5
h LBL 0	038-25, 13, 0
RCL 1	039- 24 1
−	040- 41
RCL 4	041- 24 4
×	042- 61
x⇄y	043- 21
÷	044- 71
R/S	045- 74
h LBL 6	046-25, 13, 6
ENTER↑	047- 31
+	048- 51
STO + 0	049-23, 51, 0
h RTN	050- 25 12
h LBL 7	051-25, 13, 7
2	052- 2
÷	053- 71
h FRAC	054- 25 33
g x=0	055- 15 71
h RTN	056- 25 12
0	057- 0

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[+]	058- 71	[h] [RTN]	059- 25 12

REGISTERS			I Unused
R_0 Used	R_1 $f(x_i)$, a	R_2 b	R_3 n
R_4 h	R_5 Used	R_6—$R_{.9}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the spacing between			
	x-values.	h	[A]	h
3	Repeat step 3 for j = 0, 1,			
	..., n: input the function			
	value at x_j.	$f(x_j)$	[B]	j
4	Calculate the area by trap-		[GSB] 1	*TRAP* ∫
	ezoidal rule, OR,			
	Calculate the area by		[GSB] 2	*SIMP* ∫
	Simpson's rule (n must be			
	even)			
5	Repeat step 4 if you want the			
	results again.			
6	For a new case, go to step 2.			

Example:

Given the values below for $f(x_j)$, $j = 0, 1, \ldots, 8$, calculate the approximations to the integral

$$\int_0^2 f(x)\,dx$$

by the trapezoidal rule and by Simpson's rule.

The value for h is 0.25.

i	0	1	2	3	4	5	6	7	8
x_i	0	.25	.5	.75	1	1.25	1.5	1.75	2
$f(x_i)$	2	2.8	3.8	5.2	7	9.2	12.1	15.6	20

Keystrokes:	**Display:**	
[f] [FIX] 2		
.25 [A]		
2 [B] 2.8 [B]		
3.8 [B]		
5.2 [B] 7 [B] 9.2 [B]		
12.1 [B] 15.6 [B] 20 [B]		
[GSB] 1	***16.68***	(Trapezoidal)
[GSB] 2	***16.58***	(Simpson's)

Differential Equations

This program solves first- and second-order differential equations by the fourth-order Runge-Kutta method. A first-order equation is of the form $y' = f(x, y)$, with initial values x_0, y_0; a second-order equation is of the form $y'' = f(x, y, y')$, with initial values x_0, y_0, y'_0.

In either case, the function f should be keyed into program memory under LBL0, and should assume that x_0 and y_0 are in the X- and Y-registers respectively; y'_0 will be in the Z-register for second-order equations. There are 109 lines left after the 1st order equation program and 78 lines left after the 2nd order equation program to define the function.

The solution is a numerical solution which calculates y_i for $x_i = x_0 + ih$ ($i = 1, 2, 3, \ldots$), where h is an increment specified by the user. The value for h may be changed at any time during the program's execution. This allows solution of the equation for values of x arbitrarily close to a pole ($y \rightarrow \pm\infty$), or a zero ($y \rightarrow 0$).

Equations:

1st Order

$$y_{i+1} = y_i + \frac{1}{6}(c_1 + 2c_2 + 2c_3 + c_4)$$

where

$$c_1 = hf(x_i, y_i)$$

$$c_2 = hf\left(x_i + \frac{h}{2}, y_i + \frac{c_1}{2}\right)$$

$$c_3 = hf\left(x_i + \frac{h}{2}, y_i + \frac{c_2}{2}\right)$$

$$c_4 = hf(x_i + h, y_i + c_3)$$

2nd Order

$$y_{i+1} = y_i + h\left[y'_i + \frac{1}{6}(k_1 + k_2 + k_3)\right]$$

$$y'_{i+1} = y'_i + \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$$

$$k_1 = hf(x_i, y_i, y'_i)$$

$$k_2 = hf\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}y'_i + \frac{h}{8}k_1, y'_i + \frac{k_1}{2}\right)$$

$$k_3 = hf\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}y'_i + \frac{h}{8}k_1, y'_i + \frac{k_2}{2}\right)$$

$$k_4 = hf\left(x_i + h,\ y_i + hy'_i + \frac{h}{2}k_3,\ y'_i + k_3\right)$$

Remarks:

1. When inputting values for a second-order solution, the values for x_0 and y_0 must be input before the value of y'_0. All values must be input, even if zero.

2. If the program is to be run for different functions, be sure that the previous function is no longer in program memory when the second is keyed in. Refer to the Instructions for deleting the previous function.

 1st Order

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-	[ENTER↑]	029- 31
[h] [LBL] [A]	001-25, 13, 11	[+]	030- 51
2	002- 2	[RCL] 2	031- 24 2
[÷]	003- 71	[+]	032- 51
[STO] 0	004- 23 0	[RCL] 1	033- 24 1
[g] [R↓]	005- 15 22	[RCL] 0	034- 24 0
[STO] 2	006- 23 2	[ENTER↑]	035- 31
[g] [R↓]	007- 15 22	[+]	036- 51
[STO] 1	008- 23 1	[STO] 3	037- 23 3
[R/S]	009- 74	[+]	038- 51
[h] [LBL] [B]	010-25, 13, 12	[GSB] 0	039- 13 0
[RCL] 2	011- 24 2	[RCL] 5	040- 24 5
[RCL] 1	012- 24 1	[+]	041- 51
[GSB] 0	013- 13 0	[RCL] 4	042- 24 4
[STO] 5	014- 23 5	[ENTER↑]	043- 31
[RCL] 2	015- 24 2	[+]	044- 51
[+]	016- 51	[+]	045- 51
[RCL] 1	017- 24 1	3	046- 3
[RCL] 0	018- 24 0	[÷]	047- 71
[+]	019- 51	[STO] [+] 2	048-23, 51, 2
[GSB] 0	020- 13 0	[RCL] 3	049- 24 3
[STO] 4	021- 23 4	[STO] [+] 1	050-23, 51, 1
[RCL] 2	022- 24 2	[RCL] 1	051- 24 1
[+]	023- 51	[R/S]	052- 74
[RCL] 1	024- 24 1	[RCL] 2	053- 24 2
[RCL] 0	025- 24 0	[R/S]	054- 74
[+]	026- 51	[GTO] [B]	055- 22 12
[GSB] 0	027- 13 0	[h] [LBL] 0	056-25, 13, 0
[STO] [+] 4	028-23, 51, 4	[RCL] 0	057- 24 0

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
×	058- 61	h RTN	059- 25 12

REGISTERS			I Used
R_0 $h/2$	R_1 x_i	R_2 y_i	R_3 h, y_i'
R_4 Used	R_5 Used	R_6—$R_{.9}$ Unused	

2nd Order

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	000-	STO 4	021- 23 4
h LBL A	001-25, 13, 11	STO f I	022-23, 14, 23
2	002- 2	RCL 0	023- 24 0
÷	003- 71	RCL f I	024-24, 14, 23
STO 0	004- 23 0	RCL 3	025- 24 3
g R↓	005- 15 22	+	026- 51
STO 3	006- 23 3	RCL 5	027- 24 5
g R↓	007- 15 22	GSB 7	028- 13 7
STO 2	008- 23 2	STO + 4	029-23, 51, 4
g R↓	009- 15 22	ENTER↑	030- 31
STO 1	010- 23 1	+	031- 51
R/S	011- 74	STO f I	032-23, 14, 23
h LBL B	012-25, 13, 12	RCL 0	033- 24 0
RCL 3	013- 24 3	ENTER↑	034- 31
RCL 2	014- 24 2	+	035- 51
RCL 1	015- 24 1	GSB 8	036- 13 8
GSB 0	016- 13 0	RCL 4	037- 24 4
STO 5	017- 23 5	ENTER↑	038- 31
STO f I	018-23, 14, 23	+	039- 51
RCL 0	019- 24 0	+	040- 51
GSB 8	020- 13 8	RCL 5	041- 24 5

KEY ENTRY	DISPLAY
+	042- 51
3	043- 3
÷	044- 71
RCL 3	045- 24 3
+	046- 51
STO 3	047- 23 3
h LST X	048- 25 0
RCL 4	049- 24 4
RCL 5	050- 24 5
+	051- 51
3	052- 3
÷	053- 71
+	054- 51
RCL 0	055- 24 0
ENTER↑	056- 31
+	057- 51
STO 5	058- 23 5
×	059- 61
RCL 2	060- 24 2
+	061- 51
STO 2	062- 23 2
RCL 5	063- 24 5
STO + 1	064-23, 51, 1
RCL 1	065- 24 1
R/S	066- 74

KEY ENTRY	DISPLAY
RCL 2	067- 24 2
R/S	068- 74
GTO B	069- 22 12
h LBL 8	070-25, 13, 8
RCL f I	071-24, 14, 23
RCL 3	072- 24 3
+	073- 51
RCL f I	074-24, 14, 23
h LBL 7	075-25, 13, 7
2	076- 2
÷	077- 71
RCL 3	078- 24 3
+	079- 51
f R↓	080- 14 22
×	081- 61
RCL 2	082- 24 2
+	083- 51
RCL 1	084- 24 1
f R↓	085- 14 22
+	086- 51
h LBL 0	087-25, 13, 0
RCL 0	088- 24 0
×	089- 61
h RTN	090- 25 12

REGISTERS			I Used
R_0 $h/2$	R_1 x_i	R_2 y_i	R_3 h, y_i'
R_4 Used	R_5 Used	R_6—$R_{.6}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
	First Order			
1	Key in the program.			
2	Prepare to key in function			
	$f(x, y)$ under [LBL] 0 by:			
	• Switch to RUN		[GTO] 0	
	• Switch to PRGM			056-25, 13, 0
	• Key in the function (do not			
	need RTN)			
	• Switch to RUN			
3	Input initial values x and y			
	and step size h	x_0	[ENTER↑]	
		y_0	[ENTER↑]	
		h	[A]	x_0
4	Calculate successive values of			
	x and y:		[B]	x_1
			[R/S]	y_1
			[R/S]	x_2
			[R/S]	y_2
			⋮	etc.
5	(Optional) To change the			
	value of h during calculation,	new h	[ENTER↑] 2	
			[÷] [STO] 0	(new h)/2
	then proceed for further			
	calculation.		[R/S]	
			⋮	
6	For a different function, first			
	delete the previous function			
	by →		[h] [RTN]	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
			[h] [BST] [h] [BST]	
			[h] [BST] [h] [BST]	
			Switch to PRGM	
			[h] [DEL]	
	Continue pressing [h] [DEL]		:	
	until		:	
	h LBL 0			056–25, 13, 0
	then go to step 2.			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
	Second Order			
1	Key in the program.			
2	Prepare to key in function			
	$f(x, y, y')$ under LBL 0 by:			
	• Switch to RUN		[GTO] 0	
	• Switch to PRGM			087–25,13,0
	• Key in the function (do not			
	need RTN)			
	• Switch to RUN			
3	Input initial values	x_0	[ENTER↑]	
		y_0	[ENTER↑]	
		y_0'	[ENTER↑]	
	h	[A]	x_0	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
4	Calculate successive values of			
	x and y:		[B]	x_1
			[R/S]	y_1
			[R/S]	x_2
			[R/S]	y_2
			⋮	etc.
5	(Optional) To change the			
	value of h during calculation,	new h	[ENTER↑] 2	
			[÷] [STO] 0	(new h)/2
	then proceed for further			
	calculation.		[R/S]	
			⋮	
6	For a different function, first			
	delete the previous function			
	by →		[h] [RTN]	
			[h] [BST] [h] [BST]	
			[h] [BST] [h] [BST]	
			Switch to PRGM	
			[h] [DEL]	
	Continue pressing [h] [DEL]			
	until h LBL 0		⋮	087–25,13,0
	then go to step 2.			

Example 1:

Solve numerically the first-order differential equation

$$y' = \frac{\sin x + \tan^{-1}(y/x)}{y - \ln(\sqrt{x^2 + y^2})}$$

where $x_0 = y_0 = 1$. Let $h = 0.5$. The angular mode must be set to radians.

Keystrokes:	**Display:**	
Key in the 1st order program and switch to RUN		
[GTO] 0		
Switch to PRGM	***056- 25, 13, 0***	
[g] [RAD] [STO] 6		
[x⇄y] [STO] 7 [x⇄y]		
[g] [→P] [f] [LN] [STO] 8		
[g] [R↓] [RCL] 6		
[f] [SIN] [+] [RCL] 7		
[RCL] 8 [−] [÷]		
[g] [DEG]	***073- 15 11***	
Switch to RUN		
[f] [FIX] 2		
1 [ENTER↑] 1 [ENTER↑]		
.5 [A]	***1.00***	
[B]	***1.50***	(x_1)
[R/S]	***2.06***	(y_1)
[R/S]	***2.00***	(x_2)
[R/S]	***2.78***	(y_2)
[R/S]	***2.50***	(x_3)
[R/S]	***3.28***	(y_3)
etc.		

Example 2:

Solve the second-order equation

$$(1 - x^2)y'' + xy' = x$$

where $x_0 = y_0 = y'_0 = 0$ and $h = 0.1$.

Rewrite the equation as

$$y'' = \frac{x\,(1 - y')}{1 - x^2} \quad , |x| \neq 1$$

Keystrokes:	**Display:**	
Key in the 2nd order program		
Switch to RUN		
[GTO] 0		
Switch to PRGM	***087- 25, 13, 0***	
[STO] 8		
[g] [R↓] [g] [R↓]		
1 [−]		
[RCL] 8 [×] [RCL] 8		
[g] [x^2] 1 [−] [÷]	***099- 71***	
Switch to RUN		
[f] [FIX] 4		
0 [ENTER↑] 0 [ENTER↑]		
0 [ENTER↑] .1 [A]	***0.0000***	
[B]	***0.1000***	(x_1)
[R/S]	***0.0002***	(y_1)
[R/S]	***0.2000***	(x_2)
[R/S]	***0.0013***	(y_2)
[R/S]	***0.3000***	(x_3)
[R/S]	***0.0046***	(y_3)

etc.

Vector Operations

This program calculates the basic vector operations of addition, dot (scalar) product, and cross product for three dimensional vectors. It also calculates the angle between two vectors. The program is capable of doing chain calculations whenever the result is a vector (refer to examples).

Equations:

Define a vector V in 3 dimensional rectangular coordinate system,

$$V = xi + yj + zk$$

then:

Vector addition:

$$V_1 + V_2 = (x_1 + x_2)\, i + (y_1 + y_2)\, j + (z_1 + z_2)\, k$$

Dot or scaler product:

$$V_1 \cdot V_2 = x_1\, x_2 + y_1\, y_2 + z_1\, z_2$$

Cross product:

$$V_1 \times V_2 = (y_1 z_2 - z_1 y_2)\mathrm{i} + (z_1 x_2 - x_1 z_2)\mathrm{j} + (x_1 y_2 - y_1 x_2)\mathrm{k}$$

Angle between vectors:

$$\gamma = \cos^{-1} \frac{V_1 \cdot V_2}{|V_1|\ |V_2|}$$

Remarks:

- For two dimensional vectors, simply consider that the k component does not exist, i.e. input 0 for z's.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
STO 3	002– 23 3
g R↓	003– 15 22
STO 2	004– 23 2
g R↓	005– 15 22
STO 1	006– 23 1
R/S	007– 74
h LBL B	008–25, 13, 12
STO 6	009– 23 6
g R↓	010– 15 22
STO 5	011– 23 5
g R↓	012– 15 22
STO 4	013– 23 4
R/S	014– 74
h LBL 1	015–25, 13, 1
RCL 1	016– 24 1
RCL 4	017– 24 4
+	018– 51
R/S	019– 74
RCL 2	028– 24 2
RCL 5	021– 24 5
+	022– 51
R/S	023– 74
RCL 3	024– 24 3
RCL 6	025– 24 6
+	026– 51
R/S	027– 74
h LBL 2	028–25, 13, 2

KEY ENTRY	DISPLAY
RCL 1	029– 24 1
RCL 4	030– 24 4
×	031– 61
RCL 2	032– 24 2
RCL 5	033– 24 5
×	034– 61
+	035– 51
RCL 3	036– 24 3
RCL 6	037– 24 6
×	038– 61
+	039– 51
h RTN	040– 25 12
h LBL 3	041–25, 13, 3
RCL 2	042– 24 2
RCL 6	043– 24 6
×	044– 61
RCL 3	045– 24 3
RCL 5	046– 24 5
×	047– 61
−	048– 41
R/S	049– 74
RCL 3	050– 24 3
RCL 4	051– 24 4
×	052– 61
RCL 1	053– 24 1
RCL 6	054– 24 6
×	055– 61
−	056– 41
R/S	057– 74

KEY ENTRY	DISPLAY
[RCL] 1	058- 24 1
[RCL] 5	059- 24 5
[×]	060- 61
[RCL] 2	061- 24 2
[RCL] 4	062- 24 4
[×]	063- 61
[−]	064- 41
[R/S]	065- 74
[h] [LBL] 4	066- 25, 13, 4
[GSB] 2	067- 13 2
[RCL] 1	068- 24 1
[g] [x^2]	069- 15 3
[RCL] 2	070- 24 2
[g] [x^2]	071- 15 3
[+]	072- 51
[RCL] 3	073- 24 3
[g] [x^2]	074- 15 3
[+]	075- 51
[RCL] 4	076- 24 4
[g] [x^2]	077- 15 3
[RCL] 5	078- 24 5
[g] [x^2]	079- 15 3
[+]	080- 51
[RCL] 6	081- 24 6
[g] [x^2]	082- 15 3
[+]	083- 51
[×]	084- 61
[f] [$\sqrt{x}$]	085- 14 3
[÷]	086- 71
[g] [$\cos^{-1}$]	087- 15 8
[R/S]	088- 74

REGISTERS			I
R_0	$R_1\ x_1$	$R_2\ y_1$	$R_3\ z_1$
$R_4\ x_2$	$R_5\ y_2$	$R_6\ z_2$	R_7—$R_{.6}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the 1st vector V_1.	x_1	[ENTER↑]	
		y_1	[ENTER↑]	
		z_1	[A]	x_1
3	Input the 2nd vector V_2.	x_2	[ENTER↑]	
		y_2	[ENTER↑]	
		z_2	[B]	x_2
	Vector Addition			
4	Calculate $V_1 + V_2$.		[GSB] 1	x
			[R/S]	y
			[R/S]	z
	Vector Dot Product			
5	Calculate $V_1 \cdot V_2$.		[GSB] 2	$V_1 \cdot V_2$
	Vector Cross Product			
6	Calculate $V_1 \times V_2$.		[GSB] 3	x
			[R/S]	y
			[R/S]	z
	Angle Between Two			
	Vectors			
7	Calculate γ.		[GSB] 4	γ
8	If you want the results again,			
	go to any of steps 4–step 7.			
9	To replace any of the vectors,			
	go to step 2 or step 3.			
10	For a new case, go to step 2.			

Example 1:

$V_1 = (2, 5, 2)$, $V_2 = (3, 3, -4)$
Addition: $V_1 + V_2 = (5, 8, -2)$
Dot product: $V_1 \cdot V_2 = 13.00$
Cross product: $V_1 \times V_2 = (-26, 14, -9)$
Angle between vectors: $\gamma = 67.16°$

Keystrokes:	Display:	
[f] [FIX] 2		
2 [ENTER↑] 5 [ENTER↑]		
2 [A]	2.00	
3 [ENTER↑] 3 [ENTER↑]		
4 [CHS] [B]	3.00	
[GSB] 1	5.00	(x)
[R/S]	8.00	(y)
[R/S]	-2.00	(z)
[GSB] 2	13.00	$(V_1 \cdot V_2)$
[GSB] 3	-26.00	(x)
[R/S]	14.00	(y)
[R/S]	-9.00	(z)
[GSB] 4	67.16	$(\gamma)^0$

Example 2:

Calculate $(V_1 + V_2) \cdot V_3$ for $V_1 = (1.10, 3.00, 4.40)$
$V_2 = (1.24, 2.17, 3.03)$, and $V_3 = (0.072, 0.231, 0.409)$

Keystrokes:	Display:	
[f] [FIX] 2		
1.10 [ENTER↑] 3 [ENTER↑]		
4.40 [A]	1.10	
1.24 [ENTER↑]		
2.17 [ENTER↑] 3.03 [B]	1.24	
[GSB] 1	2.34	
[R/S]	5.17	$(V_1 + V_2)$
[R/S]	7.43	
[A] 0.072 [ENTER↑]		
0.231 [ENTER↑]		
0.409 [B]	0.07	
[GSB] 2	4.40	$((V_1 + V_2) \cdot V_3)$

Triangle Solutions

This program may be used to find the sides, the angles, and the area of a plane triangle.



In general, the specification of any three of the six parameters of a triangle (3 sides, 3 angles) is sufficient to define the triangle. (The exception is that three angles will not define a triangle.) There are thus five possible cases that this program will handle: two sides and the included angle (SAS), two angles and the included side (ASA), two sides and the adjacent angle (SSA—an ambiguous case), two angles and the adjacent side (AAS), and three sides (SSS).

If the three known input values are selected in a clockwise order around the triangle the outputs will also follow a clockwise order. The outputs are obtained by reviewing the storage registers as follows:

Area	Register 0
Side 1	Register 1
Angle 1	Register 2
Side 2	Register 3
Angle 2	Register 4
Side 3	Register 5
Angle 3	Register 6

Remarks:

Inputs may be in any angular mode (i.e., DEG, RAD, GRAD). Be sure calculator is set to match the angular mode of the inputs.

- Note that the triangle described by the program does not conform to standard triangle notation; i.e., A_1 is not opposite S_1.
- Angles must be entered as decimals. The [g] [→H] conversion can be used to convert degrees, minutes, and seconds to decimal degrees.
- Accuracy of solution may degenerate for triangles containing extremely small angles.

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-
[h] [LBL] 1	001-25, 13, 1
[STO] 5	002- 23 5
[g] [R↓]	003- 15 22
[STO] 3	004- 23 3
[g] [R↓]	005- 15 22
[STO] 1	006- 23 1
[RCL] 3	007- 24 3
[g] [→P]	008- 15 4
[g] [x²]	009- 15 3
[RCL] 5	010- 24 5
[g] [x²]	011- 15 3
[−]	012- 41
[RCL] 1	013- 24 1
[RCL] 3	014- 24 3
[×]	015- 61
2	016- 2
[×]	017- 61
[÷]	018- 71
[g] [COS⁻¹]	019- 15 8
[STO] 2	020- 23 2
[GTO] 0	021- 22 0

KEY ENTRY	DISPLAY
[h] [LBL] 2	022-25, 13, 2
[STO] 2	023- 23 2
[g] [R↓]	024- 15 22
[STO] 1	025- 23 1
[g] [R↓]	026- 15 22
[STO] 6	027- 23 6
[f] [SIN]	028- 14 7
[RCL] 2	029- 24 2
[RCL] 6	030- 24 6
[+]	031- 51
[f] [SIN]	032- 14 7
[÷]	033- 71
[RCL] 1	034- 24 1
[×]	035- 61
[STO] 3	036- 23 3
[GTO] 0	037- 22 0
[h] [LBL] 3	038-25, 13, 3
[STO] 4	039- 23 4
[g] [R↓]	040- 15 22
[STO] 2	041- 23 2
[g] [R↓]	042- 15 22
[STO] 1	043- 23 1

KEY ENTRY	DISPLAY
RCL 4	044- 24 4
RCL 2	045- 24 2
+	046- 51
f SIN	047- 14 7
RCL 4	048- 24 4
f SIN	049- 14 7
÷	050- 71
RCL 1	051- 24 1
×	052- 61
STO 3	053- 23 3
GTO 0	054- 22 0
h LBL 5	055-25, 13, 5
STO 4	056- 23 4
g R↓	057- 15 22
STO 3	058- 23 3
g R↓	059- 15 22
STO 1	060- 23 1
RCL 3	061- 24 3
RCL 4	062- 24 4
f SIN	063- 14 7
RCL 1	064- 24 1
÷	065- 71
×	066- 61
g SIN⁻¹	067- 15 7
RCL 4	068- 24 4
+	069- 51
GSB 9	070- 13 9

KEY ENTRY	DISPLAY
STO 2	071- 23 2
GSB 0	072- 13 0
RCL 1	073- 24 1
RCL 3	074- 24 3
f x≤y	075- 14 41
GTO 8	076- 22 8
h LBL 7	077-25, 13, 7
2	078- 2
h PSE	079- 25 74
GTO 7	080- 22 7
h LBL A	081-25, 13, 11
RCL 6	082- 24 6
GSB 9	083- 13 9
STO 6	084- 23 6
RCL 4	085- 24 4
+	086- 51
GSB 9	087- 13 9
STO 2	088- 23 2
GTO 0	089- 22 0
h LBL 4	090-25, 13, 4
STO 3	091- 23 3
g R↓	092- 15 22
STO 2	093- 23 2
g R↓	094- 15 22
STO 1	095- 23 1
h LBL 0	096-25, 13, 0
RCL 2	097- 24 2

KEY ENTRY	DISPLAY
RCL 1	098- 24 1
f →R	099- 14 4
RCL 3	100- 24 3
x⇄y	101- 21
−	102- 41
g →P	103- 15 4
STO 5	104- 23 5
x⇄y	105- 21
STO 4	106- 23 4
RCL 2	107- 24 2
+	108- 51
GSB 9	109- 13 9
STO 6	110- 23 6
f SIN	111- 14 7
×	112- 61

KEY ENTRY	DISPLAY
RCL 1	113- 24 1
×	114- 61
2	115- 2
÷	116- 71
STO 0	117- 23 0
h RTN	118- 25 12
h LBL 9	119-25, 13, 9
f COS	120- 14 8
CHS	121- 32
g COS⁻¹	122- 15 8
h RTN	123- 25 12
h LBL 8	124-25, 13, 8
1	125- 1
h PSE	126- 25 74
GTO 8	127- 22 8

REGISTERS			I Unused
R_0 Area	R_1 S_1	R_2 A_1	R_3 S_2
R_4 A_2	R_5 S_3	R_6 A_3	R_7—$R_{.0}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Set proper angular mode and			
	desired display format.			
3	Find the applicable case and			
	input the indicated values:			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
3a	SSS (3 sides known): Side 1	S_1	[ENTER↑]	
	Side 2	S_2	[ENTER↑]	
	Side 3	S_3	[GSB] 1	Area
	Go to step 4 to review			
	solution.			
3b	ASA (2 angles and included			
	side known):			
	Angle 3	A_3	[ENTER↑]	
	Side 1	S_1	[ENTER↑]	
	Angle 1	A_1	[GSB] 2	Area
	Go to step 4 to review			
	solution.			
3c	SAA (2 angles and adjacent			
	side known):			
	Side 1	S_1	[ENTER↑]	
	Angle 1	A_1	[ENTER↑]	
	Angle 2	A_2	[GSB] 3	Area
	Go to step 4 to review			
	solution.			
3d	SAS (2 sides and included			
	side known):			
	Side 1	S_1	[ENTER↑]	
	Angle 1	A_1	[ENTER↑]	
	Side 2	S_2	[GSB] 4	Area
	Go to step 4 to review			
	solution.			
3e	SSA (2 sides and adjacent			
	angle known):			
	Side 1	S_1	[ENTER↑]	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
	Side 2	S_2	[ENTER↑]	
	Angle 2	A_2	[GSB] 5	"1" or "2"
	One or two possible solutions			
	may exist for this case:			
	1) If the display flashes "1"			
	only one solution exists. To			
	review the solution press			
	[R/S] and go to step 4.		[R/S]	1.0000
	2) If the display flashes "2",			
	two solutions exist. To			
	review the first solution,			
	press [R/S] and go to step			
	4. After reviewing the first		[R/S]	2.0000
	solution, press [A] to cal-			
	culate the 2nd solution,			
	then go to step 4 to review			
	it.		[A]	Area
4	To review solutions:			
	Area of the triangle		[RCL] 0	Area
	Side 1		[RCL] 1	S_1
	Angle 1		[RCL] 2	A_1
	Side 2		[RCL] 3	S_2
	Angle 2		[RCL] 4	A_2
	Side 3		[RCL] 5	S_3
	Angle 3		[RCL] 6	A_3

Example 1:

A surveyor is to find the area and dimensions of a triangular land parcel. From point A, the distances to B and C are measured with an electronic distance meter. The angle between AB and AC is also measured. Find the area and other dimensions of the triangle.



This is a side-angle-side problem where:

$$S_1 = 171.63, \ A_1 = 98°12' \text{ and } S_2 = 297.35.$$

Keystrokes:	Display:	
[g] [DEG] [f] [FIX] 2		
171.63 [ENTER↑]	***171.63***	
98.12 [g] [→H]	***98.20***	
297.35 [GSB] 4	***25,256.21***	Area (m^2)
[RCL] 4	***27.83***	Angle C
[RCL] 5	***363.91***	Side BC
[RCL] 6	***53.97***	Angle B

Example 2:

Given 2 sides and a nonincluded angle solve for the triangle:

Side $_1$ = 25.6
Side $_2$ = 32.8
Angle$_2$ = 42.3°



Keystrokes:	Display:		
25.6 [ENTER↑]	***25.60***		
32.8 [ENTER↑]	***32.80***		
42.3 [GSB] 5	***"2"***	2 solutions	
[R/S] [RCL] 0	***410.85***	Area	
[RCL] 1	***25.60***	S_1	
[RCL] 2	***78.12***	A_1	
[RCL] 3	***32.80***	S_2	1st solution
[RCL] 4	***42.30***	A_2	
[RCL] 5	***37.22***	S_3	
[RCL] 6	***59.58***	A_3	
[A]	***124.68***	Area	
[RCL] 1	***25.60***	S_1	
[RCL] 2	***17.28***	A_1	
[RCL] 3	***32.80***	S_2	2nd solution
[RCL] 4	***42.30***	A_2	
[RCL] 5	***11.30***	S_3	
[RCL] 6	***120.42***	A_3	

Coordinate Transformation

This program performs translation, rotation and rescaling of coordinates from one coordinate system to another. A point in the original system (x_0, y_0) becomes the origin of the new system. The x and y axis are rotated through an angle α with respect to the old coordinate system. The new coordinate system may also have a different scale, if desired. A point P having coordinates (x_p, y_p) in the old system now has coordinates (x'_p, y'_p) in the new system.



Equations:

$$\theta = \text{Tan}^{-1}\left(\frac{y_p - y_o}{x_p - x_o}\right) - \alpha$$

$$D = S\sqrt{(y_p - y_o)^2 + (x_p - x_o)^2}$$

$$x'_p = D\cos\theta$$

$$y'_p = D\sin\theta$$

where:

x_o, y_o = coordinates of the new origin in the old system

x_p, y_p = coordinates of a point P in the old system

x'_p, y'_p = coordinates of point P in the new system

Input data required are the coordinates of the new origin in the old coordinate system, the rotation angle and the scale factor (if other than 1).

Points in the original system may be converted to the transformed system using key [A]. Points in the new system may be converted to the original system using key [B].

Remarks:

- The scale factor need not be input unless it is other than one.
- Be sure the rotation angle is input as a decimal and that the calculator is set to the proper mode (i.e., DEG, RAD, GRAD).
- For pure translation, input zero for the rotation angle.
- For pure rotation, input zeros for (x_o, y_o).

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000–
[h] [LBL] 0	001– 25, 13, 0
[STO] 1	002– 23 1
[g] [R↓]	003– 15 22
[STO] 0	004– 23 0
[R/S]	005– 74
[STO] 4	006– 23 4
1	007– 1
[STO] 5	008– 23 5
[R/S]	009– 74
[h] [LBL] [A]	010– 25, 13, 11
[RCL] 1	011– 24 1
[–]	012– 41
[x⇄y]	013– 21
[RCL] 0	014– 24 0
[–]	015– 41
[g] [→P]	016– 15 4
[RCL] 5	017– 24 5
[×]	018– 61
[x⇄y]	019– 21
[RCL] 4	020– 24 4
[–]	021– 41
[x⇄y]	022– 21
[f] [→R]	023– 14 4
[R/S]	024– 74
[x⇄y]	025– 21
[R/S]	026– 74
[h] [LBL] [B]	027– 25, 13, 12
[x⇄y]	028– 21
[g] [→P]	029– 15 4
[RCL] 5	030– 24 5
[÷]	031– 71
[x⇄y]	032– 21
[RCL] 4	033– 24 4
[+]	034– 51
[x⇄y]	035– 21
[f] [→R]	036– 14 4
[RCL] 0	037– 24 0
[+]	038– 51
[R/S]	039– 74
[x⇄y]	040– 21
[RCL] 1	041– 24 1
[+]	042– 51

REGISTERS			I Unused
R_0 x_0	R_1 y_0	R_2 Unused	R_3 Unused
R_4 α	R_5 S	R_6 – $R_{.9}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Set proper angular			
	mode and desired display.			
	format.			
3	Input the coordinates of the			
	point in the old system about			
	which the transformation is to			
	be performed.	x_o	[ENTER↑]	
		Y_0	[GSB] 0	X_0
4	Input the rotation angle (posi-			
	tive if counter-clockwise.			
	negative if clockwise).	α	[R/S]	1.0000
5	Input the scale factor (if other			
	than 1).	S	[STO] 5	S
6	Transform a point from the old			
	system to the new.	X_r	[ENTER↑]	
		Y_p	[A]	X'_p
			[R/S]	Y'_p
	OR,			
7	From the new system to the			
	old.	X'_p	[ENTER↑]	
		Y'_p	[B]	X_p
			[R/S]	Y_p
8	Repeat steps 7 or 8 for other			
	points.			
9	For a new transformation go			
	to step 2.			

Example 1:

The point (7, −4) in one coordinate system is to become the origin of a new coordinate system. The new system will also be rotated 27° counterclockwise with respect to the old system. Find the new coordinates of the points (−9, 7), (−5, −4) and (6, 8). What were the former coordinates of the new points (2.7, −3.6)?

Keystrokes:	Display:
[g] [DEG]	
[f] [FIX] 4	
7 [ENTER↑]	
4 [CHS] [GSB] 0	7.0000
27 [R/S]	1.0000
9 [CHS] [ENTER↑] 7 [A]	-9.2622
[R/S]	17.0649
5 [CHS] [ENTER↑]	
4 [CHS] [A]	-10.6921
[R/S]	5.4479
6 [ENTER↑] 8 [A]	4.5569
[R/S]	11.1461
2.7 [ENTER↑]	
3.6 [CHS] [B]	11.0401
[R/S]	-5.9818

Circle Determined by Three Points

This program calculates the center (x_0, y_0) and radius (r) of a circle given three non-collinear points.

The equation of a circle is:

$$(x - x_0)^2 + (y - y_0)^2 = r^2$$

x_0 and y_0 are solved from:

$$\begin{bmatrix} (x_1 - x_3) & (y_1 - y_3) \\ (x_1 - x_2) & (y_1 - y_2) \end{bmatrix} \begin{bmatrix} x_0 \\ y_0 \end{bmatrix} = \begin{bmatrix} A \\ B \end{bmatrix}$$

where:

$$A = \tfrac{1}{2}\left[(x_1 - x_3)(x_1 + x_3) + (y_1 - y_3)(y_1 + y_3)\right]$$
$$B = \tfrac{1}{2}\left[(x_1 - x_2)(x_1 + x_2) + (y_1 - y_2)(y_1 + y_2)\right]$$

and $r = \sqrt{(x_1 - x_0)^2 + (y_1 - y_0)^2}$

Remark:

If the determinant of the system is zero, then the three points are collinear and this program is not applicable.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
STO 1	002- 23 1
STO 8	003- 23 8
x⇄y	004- 21
STO 0	005- 23 0
R/S	006- 74
h LBL B	007-25, 13, 12

KEY ENTRY	DISPLAY
2	008- 2
GSB 0	009- 13 0
R/S	010- 74
h LBL 1	011-25, 13, 1
5	012- 5
GSB 0	013- 13 0
RCL 7	014- 24 7
RCL 2	015- 24 2

KEY ENTRY	DISPLAY
[×]	016- 61
[RCL] 4	017- 24 4
[RCL] 5	018- 24 5
[×]	019- 61
[−]	020- 41
[STO] 8	021- 23 8
[R/S]	022- 74
[h] [LBL] 2	023-25, 13, 2
[RCL] 6	024- 24 6
[RCL] 2	025- 24 2
[×]	026- 61
[RCL] 5	027- 24 5
[RCL] 3	028- 24 3
[GSB] 9	029- 13 9
[R/S]	030- 74
[RCL] 7	031- 24 7
[RCL] 3	032- 24 3
[×]	033- 61
[RCL] 6	034- 24 6
[RCL] 4	035- 24 4
[GSB] 9	036- 13 9
[R/S]	037- 74
[x⇄y]	038- 21
[RCL] 0	039- 24 0
[−]	040- 41
[x⇄y]	041- 21

KEY ENTRY	DISPLAY
[RCL] 1	042- 24 1
[−]	043- 41
[g] [→P]	044- 15 4
[R/S]	045- 74
[GTO] 2	046- 22 2
[h] [LBL] 9	047-25, 13, 9
[×]	048- 61
[−]	049- 41
2	050- 2
[÷]	051- 71
[RCL] 8	052- 24 8
[÷]	053- 71
[h] [RTN]	054- 25 12
[h] [LBL] 0	055-25, 13, 0
[STO] [f] [I]	056-23, 14, 23
[g] [R↓]	057- 15 22
[ENTER↑]	058- 31
[ENTER↑]	059- 31
[RCL] 1	060- 24 1
[−]	061- 41
[CHS]	062- 32
[STO] [f] [(i)]	063-23, 14, 24
[x⇄y]	064- 21
[RCL] 1	065- 24 1
[+]	066- 51
[×]	067- 61

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[g] [ISG]	068- 15 24	[STO] [f] [(i)]	077-23, 14, 24
[h] [PSE]	069- 25 74	[x⇄y]	078- 21
[STO] [f] [(i)]	070-23, 14, 24	[RCL] 0	079- 24 0
[CLx]	071- 34	[+]	080- 51
[RCL] 0	072- 24 0	[×]	081- 61
[−]	073- 41	[g] [DSE]	082- 15 23
[CHS]	074- 32	[STO] [+] [(i)]	083-23, 51, 24
[g] [ISG]	075- 15 24	[h] [RTN]	084- 25 12
[h] [PSE]	076- 25 74		

REGISTERS			I Index
R_0 x_1	R_1 y_1	R_2 (y_1-y_2)	R_3 Used
R_4 (x_1-x_2)	R_5 (y_1-y_3)	R_6 Used	R_7 (x_1-x_3)
R_8 Det	R_9—$R_{.7}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input (x_1, y_1)	x_1	[ENTER↑]	
		y_1	[A]	x_1
3	Input (x_2, y_2)	x_2	[ENTER↑]	
		y_2	[B]	
4	Input (x_3, y_3)	x_3	[ENTER↑]	
		y_3	[GSB] 1	determinant
5	Calculate: x_0		[GSB] 2	x_0
	y_0		[R/S]	y_0
	r		[R/S]	r
6	If you want the results again,			
	repeat step 5.			
7	For a new case, go to step 2.			

Example 1:

Find the equation of the circle that goes through the three points (1, 1), (3.5, -7.6), (12, 0.8).

Solution 1:

Center = (6.45, -2.08), r = 6.26

Equation: $(x - 6.45)^2 + (y + 2.08)^2 = (6.26)^2$

Keystrokes:	Display:	
[f] [FIX] 2		
1 [ENTER↑] 1 [A]	*1.00*	
3.5 [ENTER↑]		
7.6 [CHS] [B]		
12 [ENTER↑]		
.8 [GSB] 1	*-94.10*	(Det.)
[GSB] 2	*6.45*	(x_0)
[R/S]	*-2.08*	(y_0)
[R/S]	*6.26*	(r)

Complex Operations

This program enables you to perform chained calculations with complex numbers. The four operations of complex arithmetic (+, −, ×, ÷) are provided, as well as several of the most used functions of a complex variable z; ($|z|$, $1/z$, z^n, $z^{1/n}$, and e^z). Functions and operations may be mixed in the course of a calculation to allow evaluation of expressions like $z_3/(z_1 + z_2)$, $e^{z_1 z_2}$, $|z_1 + z_2| + |z_2 - z_3|$, etc., where z_1, z_2, z_3 are complex numbers of the form $x + iy$.

Arithmetic Operations

An arithmetic operation needs two numbers upon which to operate. Both numbers must be input before the operation can be performed. Suppose that $z_1 = 2 + 3i$, $z_2 = 5 - i$, and we wish to find $z_1 - z_2$. This can be calculated by the keystrokes 2 [ENTER↑] 3 [A] 5 [ENTER↑] 1 [CHS] [A] 2 [R/S]. The result $z_3 = u + iv$ is found to be $-3 + 4i$. This result is now stored by the program in place of the second complex number z_2. A further calculation $z_3 \times z_4$ could be performed by inputting z_4 and pressing 3 [R/S] for multiplication. This type of chaining can be continued indefinitely, and functions can be interspersed with arithmetic operations (see Step 9 of Users' Instructions for this case.)

Functions

The complex functions in this program act on just one number. Thus to perform a function, you need simply to input a complex number z and then perform the appropriate function. For example, to find the reciprocal of $(2 + 3i)$, press 2 [ENTER↑] 3 [A] 2 [B]. The result is calculated as $u + iv = 0.1538 - 0.2308i$. This result is now stored in place of the original number, and further calculations will operate on this result. All complex functions operate in this same manner, with one exception: since the function $|z|$ returns a real number, its result is not stored.

Equations:

$$\text{Let } z_j = x_j + iy_j = r_j e^{i\theta_j}, j = 1,2$$

$$z = x + iy = re^{i\theta}$$

Let the result in each case be $u + iv$

$$z_1 + z_2 = u + iv = (x_1 + x_2) + i\,(y_1 + y_2)$$

$$z_1 - z_2 = u + iv = (x_1 - x_2) + i\,(y_1 - y_2)$$

$$z_1 \cdot z_2 = u + iv = r_1 \cdot r_2 \cdot e^{i(\theta_1 + \theta_2)}$$

$$z_1/z_2 = u + iv = \frac{r_1}{r_2}\, e^{i(\theta_1 - \theta_2)}$$

$$|z| = r = \sqrt{x^2 + y^2}$$

$$\frac{1}{z} = \frac{x}{r^2} - i\frac{y}{r^2}$$

$$z^n = r^n e^{in\theta} \quad n = \pm(1, 2, 3, \ldots)$$

$$z^{1/n} = r^{1/n} e^{i\left(\frac{\theta}{n} + \frac{360k}{n}\right)}, k = 0, 1, \ldots, n-1$$

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL 0	001-25, 13, 0
STO f I	002-23, 14, 23
g R↓	003- 15 22
GSB f I	044-13, 14, 23
R/S	005- 74
h LBL A	006-25, 13, 11
h CF 0	007-25, 61, 0
g DEG	008- 15 11
RCL 4	009- 24 4
STO 2	010- 23 2
g R↓	011- 15 22
STO 4	012- 23 4
g R↓	013- 15 22
RCL 5	014- 24 5
STO 3	015- 23 3
g R↓	016- 15 22
STO 5	017- 23 5
h RTN	018- 25 12
h LBL 2	019-25, 13, 2
h SF 0	020-25, 51, 0
h LBL 1	021-25, 13, 1
RCL 3	022- 24 3
RCL 5	023- 24 5
h F? 0	024-25, 71, 0
CHS	025- 32
+	026- 51
STO 5	027- 23 5
R/S	028- 74

KEY ENTRY	DISPLAY
RCL 2	029- 24 2
RCL 4	030- 24 4
h F? 0	031-25, 71, 0
CHS	032- 32
+	033- 51
STO 4	034- 23 4
h CF 0	035-25, 61, 0
h RTN	036- 25 12
h LBL 3	037-25, 13, 3
GSB 7	038- 13 7
h LBL 9	039-25, 13, 9
x≷y	040- 21
f R↓	041 14 22
+	042- 51
x≷y	043- 21
f R↓	044- 14 22
×	045- 61
h LBL 8	046-25, 13, 8
f →R	047- 14 4
STO 5	048- 23 5
R/S	049- 74
x≷y	050- 21
STO 4	051- 23 4
g DEG	052- 15 11
h RTN	053- 25 12
h LBL 4	054-25, 13, 4
GSB 7	055- 13 7
x≷y	056- 21
CHS	057- 32

KEY ENTRY	DISPLAY
[x⇄y]	058- 21
[h] [1/x]	059- 25 2
[GTO] 9	060- 22 9
[h] [LBL] [B]	061-25, 13, 12
[STO] [f] [I]	062-23, 14, 23
[g] [R↓]	063- 15 22
[GSB] [f] [I]	064-13, 14, 23
[R/S]	065- 74
[h] [LBL] 5	066-25, 13, 5
[g] [RAD]	067- 15 12
[RCL] 4	068- 24 4
[RCL] 5	069- 24 5
[g] [e^x]	070- 15 1
[GTO] 8	071- 22 8
[h] [LBL] 2	072-25, 13, 2
[GSB] 1	073- 13 1
[x⇄y]	074- 21
[CHS]	075- 32
[x⇄y]	076- 21
[h] [1/x]	077- 25 2
[GTO] 8	078- 22 8
[h] [LBL] 3	079-25, 13, 3
[STO] 0	080- 23 0
[GSB] 1	081- 13 1
[RCL] 0	082- 24 0
[h] [y^x]	083- 25 3
[x⇄y]	084- 21
[RCL] 0	085- 24 0
[×]	086- 61
[x⇄y]	087- 21
[GTO] 8	088- 22 8
[h] [LBL] 4	089-25, 13, 4
[STO] 0	090- 23 0
3	091- 3
6	092- 6
0	093- 0
[x⇄y]	094- 21
[÷]	095- 71
[STO] 1	096- 23 1
[GSB] 1	092- 13 1
[RCL] 0	098- 24 0
[h] [1/x]	099- 25 2
[h] [y^x]	100- 25 3
[x⇄y]	101- 21
[RCL] 0	102- 24 0
[÷]	103- 71
[x⇄y]	104- 21
[RCL] 0	105- 24 0
[STO] [f] [I]	106-23, 14, 23
[g] [R↓]	107- 15 22
[h] [LBL] 6	108-25, 13, 6
[GSB] 8	109- 13 8
[R/S]	110- 74
[g] [DSE]	111- 15 23
[GTO] 5	112- 22 5
[h] [RTN]	113- 25 12
[h] [LBL] 5	114-25, 13, 5
[x⇄y]	115- 21

KEY ENTRY	DISPLAY
[g] [→P]	116- 15 4
[x⇄y]	117- 21
[RCL] 1	118- 24 1
[+]	119- 51
[x⇄y]	120- 21
[GTO] 6	121- 22 6
[h] [LBL] 7	122-25, 13, 7

KEY ENTRY	DISPLAY
[RCL] 2	123- 24 2
[RCL] 3	124- 24 3
[g] [→P]	125- 15 4
[h] [LBL] 1	126-25, 13, 1
[RCL] 4	127- 24 4
[RCL] 5	128- 24 5
[g] [→P]	129- 15 4

REGISTERS			I Code
R_0 n	R_1	$R_2 y_1$	$R_3 x_1$
$R_4 y_2$	$R_5 x_2$	R_6—$R_{.0}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Key in the first complex	x_1	[ENTER↑]	
	number, $z_1 = x_1 + iy_1$.	y_1	[A]	x_1
3	For a function, go to step 7. For			
	arithmetic, go to step 4. The			
	complex result is $u + iv$.			
4	ARITHMETIC:			
	Key in the second complex	x_2	[ENTER↑]	
	number, $z_2 = x_2 + iy_2$.	y_2	[A]	x_2
5	Select one of the four arith-			
	metic operations by keying in			
	the code:			
	• Add, (+) (code = 1)	1	[R/S]	u
			[R/S]	v
	• Subtract, (−) (code = 2)	2	[R/S]	u

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
			[R/S]	v
	• Multiply, (×) (code = 3)	3	[R/S]	u
			[R/S]	v
	• Divide, (÷) (code = 4)	4	[R/S]	u
			[R/S]	v
6	The result of the operation has			
	been stored. Go to step 7 for a			
	function or return to step 4 for			
	further arithmetic.			
7	FUNCTIONS:			
	Select one of the five functions:			
	• Magnitude, ($\|z\|$) (code = 1)	1	[B]	$\|z\|$
	• Reciprocal, ($1/z$) (code = 2)	2	[B]	u
			[R/S]	v
	• Raise to an integer power,			
	(z^n) (code = 3)	n	[ENTER↑]	
		3	[B]	u
			[R/S]	v
	• Find the n[th] root of z, ($z^{1/n}$)			
	(code = 4)	n	[ENTER↑]	
		4	[B]	u_1
	NOTE: n roots ($u + iv$) will be		[R/S]	v_1
	found.		[R/S]	u_2
	Continue to press [R/S] until		[R/S]	v_2
	all the roots are found.		⋮	⋮
			[R/S]	u_n
			[R/S]	v_n
	• Raise e to the power z, (e^z)	5	[B]	u
	(code = 5)		[R/S]	v

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
8	The result, if complex, has been			
	stored; go to step 4 for arith-			
	metic or step 7 for another			
	function.			
9	To perform arithmetic on two			
	complex numbers which are			
	already stored in the calculator			
	(such as after performing			
	functions), key in the code for			
	the desired arithmetic operation			
	as in step 4 and press [GSB] 0.	code	[GSB] 0	*u*
			[R/S]	*v*

Examples:

Keystrokes: **Display:**

1. $(3 + 4i) + (7.4 - 5.6i) = 10.40 - 1.60i$

[f] [FIX] 4
3 [ENTER↑] 4 [A] 7.4
[ENTER↑] 5.6 [CHS] [A]
1 [R/S] **10.4000**
[R/S] **-1.6000**

2. $(3 + 4i) - (7.4 - 5.6i) = -4.40 + 9.60i$

3 [ENTER↑] 4 [A] 7.4
[ENTER↑] 5.6 [CHS] [A]
2 [R/S] **-4.4000**
[R/S] **9.6000**

3. $(3.1 + 4.6i) \times (5 - 12i) = 70.70 - 14.20i$

3.1 [ENTER↑] 4.6 [A]
5 [ENTER↑] 12 [CHS] [A]

Keystrokes:

3 [R/S]	***70.7000***	
[R/S]	***-14.2000***	

4. $\dfrac{3 + 4i}{7 - 2i}$ $0.2453 + 0.6415i$

3 [ENTER↑] 4 [A]		
7 [ENTER↑] 2 [CHS] [A]		
4 [R/S]	***0.2453***	
[R/S]	***0.6415***	

5. $\dfrac{1}{2 + 3i} = 0.1538 - 0.2308i$

2 [ENTER↑] 3 [A] 2 [B]	***0.1538***	(z^{-1})
[R/S]	***-0.2308***	

6. Find the 3 cube roots of 8.

8 [ENTER↑] 0 [A]	***8.0000***	
3 [ENTER↑] 4 [B]	***2.0000***	u_1
[R/S]	***0.0000***	v_1
[R/S]	***-1.0000***	u_2
[R/S]	***1.7321***	v_2
[R/S]	***-1.0000***	u_3
[R/S]	***-1.7321***	v_3

7. Evaluate $e^{z^{-2}}$ where $z = (1 + i)$

1 [ENTER↑] 1 [A]	***1.0000***	
2 [ENTER↑] 3 [B]	***0.0000***	(z^2)
[R/S]	***2.0000***	
2 [B]	***0.0000***	(z^{-2})
[R/S]	***-0.5000***	
5 [B]	***0.8776***	$(e^{z^{-2}})$
[R/S]	***-0.4794***	

8. $\dfrac{23 + 13i}{(-2 + i) + (4 - 3i)} = 2.5000 + 9.0000i$

2 [CHS] [ENTER↑] 1 [A]	**-2.0000**	
4 [ENTER↑] 3 [CHS] [A]		
1 [R/S]	**2.0000**	
[R/S]	**-2.0000**	
2 [B]	**0.2500**	Reciprocal
[R/S]	**0.2500**	
23 [ENTER↑] 13 [A]		
3 [R/S]	**2.5000**	Multiply
[R/S]	**9.0000**	

Base Conversions

This program will convert a positive number in base b, x_b, to its equivalent representation in base B,x_B. The bases b and B may take on integer values from 2 to 99, inclusive. Inputs to the program are x_b, b, and B; the single output is the value of x_B. If several conversions are to be done between the same two bases, i.e., there are several values of x_b for the same b and B, then the bases need not be re-input each time. Once the new value of x_b is keyed in, then pressing [A] will immediately cause the calculation of x_B, based on the most recent values for b and B.

The heart of this program is a routine which actually converts numbers to and from base 10 representations. If either b or B is equal to 10, this routine is executed just once, and then the program halts displaying x_B. If, on the other hand, neither b nor B is 10, then x_b is first converted to its decimal representation, x_{10}, then x_{10} is converted to x_B. In this case the routine is executed twice.

A number such as $4B6_{16}$ cannot be represented directly on the display because the display is strictly numeric. Therefore, some convention must be adopted to represent numbers x_a when $a > 10$. We use the convention of allocating two digit locations for each single character in x_a when $a > 10$.

For example, $4B6_{16}$ is represented as 041106_{16} by our convention (in hexadecimal system, A = 10, B = 11, C = 12, D = 13, E = 14, F = 15).

When displayed, this number may appear as 41106 or with an exponent

4.1106 04

which is interpreted as $4.B6 \times 16^2$.

The displayed exponent 4 is for base 10 and only serves to locate the decimal point (in the same manner as for decimal numbers).

When base $a > 10$ (as in the above example), divide the displayed exponent by 2 to get the true exponent of the number. When the displayed exponent is an odd integer, shift the decimal point of the displayed number one place (to the left or right) and adjust its exponent accordingly to make the true exponent an integer.

For example, the displayed number

1.112 -03

is interpreted as B.C $\times$ 16^{-2} or 0.BC $\times$ 16^{-1}.

Remarks:

1. When the magnitude of the number is very large or very small, this program will take a long time to execute.
2. The program will not give any error indication for invalid inputs for x_b. For example, 981_8 will be treated the same as 1201_8.

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-
[h] [LBL] 0	001-25, 13, 0
1	002- 1
0	003- 0
[STO] 8	004- 23 8
[g] [R↓]	005- 15 22
[STO] [•] 0	006- 23 .0
[g] [R↓]	007- 15 22
[STO] 9	008- 23 9
[R/S]	009- 74
[h] [LBL] [B]	010-25, 13, 12
[h] [CF] 0	011- 25, 61, 0
[STO] 3	012- 23 3
[RCL] 9	013- 24 9
[STO] 2	014- 23 2
[RCL] [•] 0	015- 24 .0
[GTO] 2	016- 22 2
[h] [LBL] [A]	017-25, 13, 11
[h] [CF] 0	018- 25, 61, 0
[STO] 3	019- 23 3
[RCL] [•] 0	020- 24 .0
[STO] 2	021- 23 2
[RCL] 9	022- 24 9
[h] [LBL] 2	023-25, 13, 2
[STO] 1	024- 23 1
[GSB] 5	025- 13 5
[h] [F?] 0	026-25, 71, 0
[GTO] 3	027- 22 3
[RCL] 2	028- 24 2

KEY ENTRY	DISPLAY
[GSB] 5	029- 13 5
[h] [F?] 0	030-25, 71, 0
[GTO] 3	031- 22 3
[RCL] 2	032- 24 2
[STO] [f] [I]	033-23, 14, 23
[RCL] 1	034- 24 1
[RCL] 8	035- 24 8
[f] [x≶y]	036- 14 41
[g] [x^2]	037- 15 3
[STO] 2	038- 23 2
[GSB] 3	039- 13 3
[STO] 3	040- 23 3
[RCL] [f] [I]	041-24, 14, 23
[STO] 2	042- 23 2
[RCL] 8	043- 24 8
[f] [x≶y]	044- 14 41
[g] [x^2]	045- 15 3
[STO] 1	046- 23 1
[h] [LBL] 3	047- 25, 13, 3
[RCL] 3	048- 24 3
[RCL] 1	049- 24 1
[STO] 5	050- 23 5
[RCL] 2	051- 24 2
[STO] 6	052- 23 6
0	053- 0
[STO] 0	054- 23 0
[STO] 4	055- 23 4
[EEX]	056- 33
1	057- 1

KEY ENTRY	DISPLAY
2	058- 2
STO 7	059- 23 7
RCL 3	060- 24 3
h LBL 9	061-25, 13, 9
1	062- 1
f x>y	063- 14 51
GTO 8	064- 22 8
STO + 0	065-23, 51, 0
CLX	066- 34
RCL 6	067- 24 6
÷	068- 71
STO 3	069- 23 3
GTO 9	070- 22 9
h LBL 8	071-25, 13, 8
RCL 6	072- 24 6
RCL 3	073- 24 3
×	074- 61
STO 3	075- 23 3
GSB 7	076- 13 7
RCL 4	077- 24 4
RCL 5	078- 24 5
×	079- 61
+	080- 51
STO 4	081- 23 4
RCL 3	082- 24 3
GSB 7	083- 13 7
RCL 3	084- 24 3
−	085- 41
h ABS	086- 25 34
STO 3	087- 23 3
1	088- 1
STO − 0	089-23, 41, 0
RCL 4	090- 24 4
RCL 7	091- 24 7
f x≤y	092- 14 41
GTO 6	093- 22 6
RCL 3	094- 24 3
g x≠0	095- 15 61
GTO 8	096- 22 8
h LBL 6	097-25, 13, 6
RCL 5	098- 24 5
RCL 0	099- 24 0
h y^x	100- 25 3
RCL 4	101- 24 4
×	102- 61
STO 4	103- 23 4
h RTN	104- 25 12
h LBL 7	105-25, 13, 7
EEX	106- 33
4	107- 4
+	108- 51
EEX	109- 33
4	110- 4
−	111- 41
h INT	112- 25 32
h RTN	113- 25 12
h LBL 5	114-25, 13, 5
RCL 8	115- 24 8

KEY ENTRY	DISPLAY
f x=y	116- 14 71
h SF 0	117-25, 51, 00
g x^2	118- 15 3

KEY ENTRY	DISPLAY
f x=y	119- 14 71
h SF 0	120-25, 51, 0
h RTN	121- 25 12

REGISTERS			I Used
R_0 Used	R_1 b	R_2 B	R_3 x_b
R_4 Used	R_5 b	R_6 B	R_7 10^{12}
R_8 10	R_9 6	$R_{.0}$ B	$R_{.1}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the first base b, and the	b	ENTER↑	
	second base B.	B	GSB 0	b
3	Input x_b to calculate x_B.	x_b	A	x_B
4	(Optional) Input x_B to			
	calculate x_b.	x_B	B	x_b
5	For a different x_b or x_B with			
	the same bases, go to step 3			
	or step 4.			
6	For conversion with different			
	bases, go to step 2.			

Example 1:

Find the number in base 10 of the octal ($b = 8$) number 177735. Find the number in octal of the number 65535 in base 10. Convert 0.2937 to base 8 representation.

Keystrokes:	Display:	
[f] [FIX] 0		
8 [ENTER↑] 10 [GSB] 0	**8.**	
177735 [A]	**65,501.**	
65535 [B]	**177,777.**	
[f] [FIX] 9		
.2937 [B]	**0.226277543**	(base 8)

Example 2:

Within the accuracy of the display find the ten-digit binary representation of π. ($x_b = 3.141592654$, $b = 10$, $B = 2$).

Keystrokes:	Display:
[f] [FIX] 9	
10 [ENTER↑] 2 [GSB] 0	**10.00000000**
[h] [π] [A]	**11.00100100**

Example 3:

Convert the following octal numbers ($b = 8$) into hexadecimal ($B = 16$): 7.200067×8^{-10}, 1.51356177×8^{17}.

Keystrokes:	Display:	
[f] [FIX] 9		
8 [ENTER↑] 16 [GSB] 0	**8.000000000**	
7.200067 [EEX] [CHS]		
10 [A]	**1.130000–14**	($1.D00 \times 16^{-7}$)
1.51356177 [EEX]		
17 [A]	**1.302141 25**	($13.02141\ 24 = D.2EA \times 16^{12}$)

HEWLETT PACKARD

1000 N.E. Circle Blvd., Corvallis, OR 97330

For additional sales and service information contact your local Hewlett-Packard Sales Office or Call 800/648-4711. Excludes Alaska and Hawaii. (In Nevada call 800/992-5710.)

00034-90032-1K-1.81-IPC

Printed in Singapore